The International Journal of Nautical Archaeology and Underwater Exploration (1988) **17.2**: 119–131

The application of a least squares adjustment program to underwater survey

Karen Atkinson
c/o Department of Maritime Archaeology, Western Australian Maritime Museum, Fremantle, Western Australia

Alec Duncan
Centre for Marine Science and Technology, Curtin University of Technology, Bentley, Western Australia

and Jeremy Green
Department of Maritime Archaeology, Western Australian Maritime Museum, Fremantle, Western Australia

Introduction

Trilateration is a common method of archaeological surveying of a wreck site. Using this system, positions on the site are located by measuring the distance from three (although often only two) reference points. These measurements are then used to map the stations or positions onto a site plan. Traditionally, the mapping process has been done graphically using a beam compass. A problem occurs if for some reason one of the three readings is inaccurate. Although this is obvious from the resultant 'cocked-hat', it is not clear which reading is at fault.

In terrestrial surveying a more rigorous approach is taken. Usually the surveyor carries out a survey by measuring the points of interest and incorporating these in some form of network of measured distances and angles. A number of numerical techniques are possible for the analysis of this type of survey work, the most obvious being the straightforward application of the geometrical equations that represent the graphical plotting procedure. However, this can become cumbersome when applied to complicated geometries, and often has to be tailored specifically to an individual survey. A technique which avoids these problems and has the additional advantage of providing a statistically rigorous treatment of the unavoidable errors that occur in any survey is the so-called 'least squares' technique.

The application of this technique has been developed and applied here to underwater archaeological survey work. One of the authors (Duncan) developed the program for use with an underwater sonar position fixing system, part of an Australian Research Grants Scheme project. The program was then modified for application to underwater archaeological survey and first used in Thailand on a survey of a wreck site. Because of the encouraging results and the usefulness of the program, a number of tests were run by another author (Atkinson), using the program to investigate its application in hypothetical situations. The results of all this work are incorporated in this paper.

Least squares applied to surveying

Assume a two-dimensional survey site with perimeter and stations or positions marked ready for survey (Fig. 1). The distance between each pair of stations is measured and these measurements form the raw data set. The measured distance between, for example, station i and station j is designated by R_{ij}. If station i is assumed to be at co-ordinates x_i, y_i, and station j at co-ordinates x_j, y_j, an inter-station distance can be calculated using plane geometry:

0305–7745/88/020119 + 13 $03.00/0



Figure 1. Sample two-dimensional site with perimeter and stations marked ready for survey.

$$r_{ij}=\sqrt{(x_i-x_j)^2+(y_i-y_j)^2}.$$

Unless the station co-ordinates are known exactly and the measurement process is perfect, the calculated distance r_{ij} will be different from the measured distance R_{ij}. The difference between the measured and calculated distances is termed the residual, and is given by:

$$v_{ij}=(R_{ij}-r_{ij}).$$

The least squares procedure is to find the X and Y co-ordinates of all stations such that the sum of the squares of the residuals

$$\sum_{i,j} v_{i,j}^2 \quad \text{is a minimum.}$$

In practice, some distance measurements are likely to be more reliable than others, and this can be taken into account by assigning a weight, w_{ij} to each measurement. The weight for a particular measurement is usually taken as the inverse of the expected variance of the measurement (in other words, the variance that would be obtained if that distance was measured a large number of times). When weights are included, it is necessary to minimize:

$$\sum_{i,j} w_{ij}\, v_{ij}^2.$$

The mathematical procedure for performing the minimization is complicated by the need to use matrix algebra and the necessary equations are presented in Appendix A. The procedure is iterative, yielding progressively more accurate results, and is halted when the desired accuracy has been obtained. A description of the algorithm is given in Appendix B and the resultant Basic program is given in Appendix C.

The least squares procedure is most useful when more distances have been measured than the absolute minimum necessary to define the station co-ordinates. If only the minimum necessary number of distances have been measured, the procedure will be able to find X and Y co-ordinates that exactly fit these measured distances, even if one or more of the distances is in error; this will result in an undetected error in the calculated co-ordinates. The minimum number of distances required to define the co-ordinates of k stations is $2k$-3.

Practical implementation

The least squares procedure has been implemented on two different systems: in Fortran IV programming language on a Cromemco CSZ computer and in Microsoft Basic on an Apple Macintosh computer. In both cases, data is input from a file which contains the measured distances, initial estimates of the station co-ordinates and an estimate of the standard deviation of each distance measurement. A set order is required so that the distance measurements can be correctly related to particular stations. Any inter-station distance that is not measured is given an approximate value and assigned a large standard deviation so that it will be given little weight in the calculation.

The initial estimates of the co-ordinates are not critical, although in order to define the origin and orientation of the grid, the *X* and *Y* co-ordinates of the first station and the *X* co-ordinate of the second station are regarded as fixed. As a consequence, the line joining the first and second stations must be made non-parallel to the *X* axis, otherwise the orientation is poorly defined and the program will fail to converge.

The output of the program consists of the following:

1. Calculated *X* and *Y* co-ordinates for each station.
2. The calculated standard deviation of the *X* and Y co-ordinates for each station. Note that these values are based on the estimates of the standard deviations of the distance measurements given as input to the program, and are only as good as those estimates.
3. A quantity called the standard error, which is a measure of how well the individual distance measurements fit together. If the estimated standard deviations in the range measurements are accurate and there are no gross measurement errors, the standard error should have a value close to one.
4. The final value of the residual for each distance measurement. The residuals indicate how much adjustment it has been necessary to make to each measured distance in order to make the co-ordinates fit, and as such provide a very useful means of locating gross errors in the measurements. Any distance measurement that corresponds to a large residual is immediately suspect.

The data output are set out so that the measurements and the resultant residuals form two similar matrices; this makes it easy to relate the measurements to the residuals, and to identify the particular distance measurement at fault.

Application and evaluation

The program was tested on an Apple Macintosh computer using 10 sets of data. Aspects of the data files were altered to determine their effect on the residuals, standard error and, most importantly, on the accuracy of the co-ordinates derived.

Firstly, an hypothetical 'wreck site', covering an area 300 mm (in the *X* direction) by 170 mm (in the *Y* direction) and comprising 10 stations, was plotted on graph paper. The distances between stations were measured to an accuracy of $\pm 0{\cdot}5$ mm and the station co-ordinates were noted.

The first data set was used as a control, with all distances and co-ordinates given to the highest level of accuracy ($\pm 0{\cdot}5$ mm). This corresponds to the ideal situation and resulted in a matrix of measurements (see Table 1, test 1). The number of ranges measured was then reduced to test the effect on the accuracy of the co-ordinate prediction. This corresponds to the situation where time is limited and the number of ranges measured has to be reduced. The ranges measured must then be selected to give an even



Figure 2. Stations plotted and joined to form triangles used to determine the minimum number of ranges acceptable and the coverage.

site coverage and the best accuracy for the situation. To obtain the best coverage, measurements were selected to form a series of linked triangles (Fig. 2). Table 1 shows the matrix formed by:

1. measuring all ranges,
2. measuring ranges derived by joining stations to form triangles, and
3. ranges measured in (2) plus chosen distances.

The ranges that were not measured were estimated and entered in the calculation with a high standard deviation.

The results obtained and tabulated in Table 2 indicate reduced co-ordinate accuracies with reduced numbers of ranges measured, compared with the control. The number of ranges measured will therefore depend on the level of accuracy required. Using all the ranges, the stations have an average standard deviation of 0·351 mm in the *X*, and 1·188 mm in the *Y* co-ordinates. Using the minimum number of ranges, the standard deviations were 0·672 mm in the *X* and 2·052 mm in the *Y* co-ordinates. By adding selected ranges to this minimum, the accuracy is increased, giving a standard deviation of 0·441 mm in *X* and 1·494 mm in the *Y* co-ordinate. It must be realized, though, that by reducing the number of ranges, errors in measurements become less easily detected, resulting in inaccurate co-ordinates.

The test site was purposely selected with a

Table 1. *Matrices of measurements used in tests*

	Station								
Station	2	3	4	5	6	7	8	9	10
Test 1: All ranges measured (measured in mm)									
1	87·70	103·65	148·35	192·95	216·35	267·00	286·75	110·95	268·52
2		67·15	74·60	123·68	133·89	186·36	217·73	75·15	180·95
3			59·12	92·63	129·74	172·65	184·22	141·45	205·35
4				49·10	71·50	118·62	143·50	141·30	148·22
5					54·35	82·05	95·00	188·95	147·10
6						53·75	101·65	182·85	93·30
7							62·70	86·30	107·55
8								278·20	170·01
9									195·00
Test 2: Measured ranges (in mm) derived by joining stations to form triangles									
1	87·7	103·65						110·95	
2		67·15	74·60		133·89			75·15	180·95
3			59·12	92·63					
4				49·1	71·5				
5					54·35		95·00		
6						53·75	101·65		93·30
7							62·70		107·55
8									
9									195·00
Test 3: Ranges measured (in mm) as in test 2 plus chosen distances									
1	87·7	103·65	148·35				286·75	110·95	268·52
2		67·15	74·60		133·89	186·36		75·15	180·95
3			59·12	92·63	129·74				205·35
4				49·1	71·5		143·5		148·22
5					54·35	82·05	95·00		
6						53·75	101·65	182·85	93·30
7							62·70	86·30	107·55
8									
9									195·00

Table 2. *Results of varying the number of ranges measured*

Station and co-ordinate	Test 1 Control	Test 2	Test 3
1 *X*	2·200	2·200	2·200
Y	14·600	14·600	14·600
2 *X*	10·200	10·200	10·200
Y	11·006	11·037	11·005
3 *X*	12·206	12·172	12·193
Y	17·411	17·456	17·412
4 *X*	17·018	17·004	17·015
Y	13·993	14·061	14·000
5 *X*	21·413	21·373	21·396
Y	16·188	16·275	16·203
6 *X*	23·594	23·592	23·592
Y	11·215	11·305	11·226
7 *X*	28·795	28·784	28·780
Y	12·592	12·695	12·612
8 *X*	30·598	30·598	30·593
Y	18·586	18·710	18·609
9 *X*	6·627	6·644	6·616
Y	4·415	4·425	4·409
10 *X*	26·011	26·035	26·009
Y	2·209	2·298	2·215
Standard error	0·328	0·176	0·219

large *X* range and a small *Y* range, a situation common on a shipwreck site. Tests were made to see what the effects of establishing two additional stations off the site were on the accuracy of the system. Two stations, 9 and 10, were added to the control data. Without these two stations, the site has a predominant spread in the *X*-axis, with little *Y*-axis component. The resulting co-ordinates are more accurate in the *Y*-axis for the test including stations 9 and 10, as is to be expected with a more even distribution. To eliminate the possibility that the increased accuracy of the co-ordinates 1–8 was due to having the two extra stations and not due to their positions, the results were compared with a test file with stations 9 and 10 located in the *X* direction.

The least squares system has the major advantage of assigning a high residual value to any measurement which has required excessive adjustment to make it fit the co-ordinates, thus indicating which of the measurements are likely to have been measured inaccurately. These distances can then be re-referenced or alternatively, assigned a high standard deviation to indicate their unreliability. The effect of assigning high standard deviations to unreliable measurements was tested in the following way.

1. With all good measurements and normal standard deviations of 0·5 mm.
2. With one inaccurate measurement, but with a normal standard deviation, i.e. the situation where the particular measurement is thought to be accurate.
3. A test with one inaccurate measurement, given a high standard deviation of 50 mm, i.e. a previous high residual has indicated this measurement to be bad.
4. Repeating (2) and (3) with three inaccurate measurements instead of one.

The results (tabulated in Table 3), indicate that measurements which are known to be unreliable, and are therefore assigned high standard deviations, will produce co-ordinates as accurate as the control test, with a similar standard error. Where bad measurements are used and assigned the normal low standard deviation, not only do the associated residuals indicate a specific problem, but the high standard error (6·622), indicates a poor reliability in the standard deviations. This gives the surveyor the information necessary to detect and correct erroneous measurements and is one of the most significant and powerful aspects of this approach.

Thailand expedition

This program was used on the preliminary survey of the Ko Si Chang II site as part of the 1987 Thai Australian expedition. The site was an ideal test situation, as the sea bed was flat with little slope or vertical structure. The site was about 30 m deep, with the associated time constraints, and low visibility required a fast survey technique, without compromising accuracy.

The site consisted of a layer of silt covering vessel timbers, some large intact jars buried to their necks and numerous ceramic sherds. Major features of the hull structure were marked with numbered tags. A total of 15 stations covered an area of 14 m E–W by 3 m N–S. All the ranges of the matrix were recorded. The measurements

Table 3. *Results of assigning varying standard deviations*

Station and co-ordinate	Control	Normal standard deviation		High standard deviation	
	All good measurements	1 Bad measurement	3 Bad measurements	1 Bad measurement	3 Bad measurements
1 *X*	2·200	2·200	2·200	2·200	2·200
Y	14·600	14·600	14·600	14·600	14·600
2 *X*	10·200	10·200	10·200	10·200	10·200
Y	11·006	10·696	10·756	11·009	11·009
3 *X*	12·206	12·288	12·304	12·205	12·205
Y	17·411	17·091	17·181	17·415	17·415
4 *X*	17·018	16·989	16·886	17·019	17·019
Y	13·993	13·538	13·377	13·998	13·999
5 *X*	21·413	21·472	21·469	21·412	21·412
Y	16·188	15·551	15·710	16·196	16·196
6 *X*	23·594	23·324	23·326	23·597	23·597
Y	11·215	10·461	10·671	11·224	11·224
7 *X*	28·795	28·702	28·654	28·796	28·796
Y	12·592	11·650	11·164	12·603	12·599
8 *X*	30·598	30·742	30·737	30·596	30·597
Y	18·586	17·558	17·658	18·597	18·596
9 *X*	6·627	6·249	6·401	6·631	6·631
Y	4·415	4·294	4·377	4·417	4·416
10 *X*	26·011	25·569	25·969	26·016	26·016
Y	2·209	1·389	1·881	2·218	2·220
Standard error	0·328	2·854	6·622	0·329	0·339

were initially taken by ten divers, but the numbers were later limited to four as inconsistencies in observers became apparent. This problem was compounded by having a number of tapes in use, some of which had both metric and Imperial units, adding reading problems caused by the effect of depth.

Initially, all the measurements were entered with standard deviations of 0·005 m. Station co-ordinates were estimated by using a rough plot of the site. High residual values from the output of the data indicated which ranges were suspect and needed to be re-measured. Re-measurement was continued until time restrained further surveying. Large standard deviations were then assigned to those few remaining unreliable measurements.

In the first test with real data, the estimates were very rough and the measurements were given equal reliability, requiring 10–15 iterations for the resulting residuals and took over an hour of computer time, before problem measurements could be identified. The program has now been altered to list residuals for each measurement on every iteration, allowing early detection of bad measurements or inaccurate estimates of standard deviation.

As our measurements were refined and estimates of the co-ordinates became more accurate, the resulting standard deviations for the co-ordinates and the residual values became uniformly small. However, the standard error, having initially been as high as 50, remained around 15. This indicated inaccurate standard deviation estimates for the distance measurements. The standard error was brought closer to one (1·5), by increasing the standard deviations from 0·005 m to 0·05 m. At this stage, the X and Y co-ordinates calculated could be confidently plotted as accurate.

The suitability of the program to different survey sites will depend largely on the environmental conditions found there. Tape measurement requires specific conditions: little water movement; reasonable visibility; and no vertical obstructions. This program is suited to obtaining positions of stations spread over a site. From this basic survey of a number of controls or reference stations, more detailed survey work can be carried out. Thus, this system would be particularly useful to establish the basic survey of control for a photomosaic. The control points can be plotted at an appropriate scale and the mosaic then scaled and adjusted to fit this. Additionally, the program is extremely useful for identifying erroneous measurements. However, where sites have variable or considerable relief, this program would not be suitable as it stands.

On sites which have a gradual slope, the third dimension (Z) can be taken into account by correcting distances first, using the Pythagoras theorem, and then entering the corrected distances in the form of co-ordinates into the program.

Conclusion

Use of this program allows easier and more accurate plotting. Its major benefit lies in the rigorous treatment of inconsistencies and discordances in observed readings. It has the advantage that it does not require artificial survey or reference stations to be established off the site and all measurements are confined within the area of interest. The program as it stands is only suitable for sites with a moderate vertical component. An electronic distance-measuring device, used in conjunction with the program, would prove to be an extremely fast and accurate method of three-dimensional underwater survey. This would do away with the inaccuracies inherent in using a tape measure.

Appendix A. Least squares adjustment program

The equations necessary to implement the least squares adjustment algorithm are stated here without proof. Details of their derivation may be found in Cross and Walker (1984), and Mikhail (1976).

Let the number of stations be denoted by k, and let m be the maximum possible number of inter-station distances, which can be shown to be: $m=k(k-1)/2$. The minimum number of distance measurements required to uniquely define the station co-ordinates is: $n=2k-3$. Using the notation of matrix algebra, the equation used to calculate the improved estimate of the station co-ordinates is:

$$\boldsymbol{X}=(\mathbf{B}^{\mathrm{T}}\mathbf{W}\mathbf{B})^{-1}\mathbf{B}^{\mathrm{T}}\mathbf{W}\boldsymbol{F}. \qquad \text{(A1)}$$

$\boldsymbol{X}$ is an n element vector of adjustments to be made to the station co-ordinates:

$$\boldsymbol{X}=[\delta y_2 \delta x_3 \delta y_3 \delta x_4 \delta y_4 \ldots\ldots \delta x_k \delta y_k]^{\mathrm{T}}.$$

$\boldsymbol{F}$ is an m element vector:

$$\boldsymbol{F}=\begin{bmatrix} R_{1,2} - r_{1,2} \\ R_{1,3} - r_{1,3} \\ \vdots \\ R_{1,k} - r_{1,k} \\ \vdots \\ R_{2,3} - r_{2,3} \\ \vdots \\ R_{2,k} - r_{2,k} \\ \vdots \\ R_{k-1,k} - r_{k-1,k} \end{bmatrix}$$

where:

$R_{i,j}$ is the measured distance between station i and station j;

$$r_{i,j}=\sqrt{(x_i-x_j)^2+(y_i-y_j)^2};$$

x_i, y_i and x_j, y_j are the assumed X, Y co-ordinates of stations i and j, respectively.

$$
\mathbf{B} = \begin{bmatrix}
-e_{1,2} & 0 & 0 & \cdot & \cdot & \cdot & 0 & 0 \\
0 & -d_{1,3} & -e_{1,3} & \cdot & \cdot & \cdot & \cdot & 0 \\
\cdot & \cdot & \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \\
0 & 0 & 0 & \cdot & \cdot & 0 & -d_{1,k} & -e_{1,k} \\
e_{2,3} & -d_{2,3} & -e_{2,3} & 0 & \cdot & \cdot & \cdot & 0 \\
e_{2,4} & 0 & 0 & -d_{2,4} & -e_{2,4} & 0 & \cdot & 0 \\
\cdot & \cdot & \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \\
e_{2,k} & 0 & \cdot & \cdot & \cdot & 0 & -d_{2,k} & -e_{2,k} \\
0 & d_{3,4} & e_{3,4} & -d_{3,4} & -e_{3,4} & 0 & \cdot & 0 \\
\cdot & \cdot & \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \\
0 & d_{3,k} & e_{3,k} & 0 & \cdot & 0 & -d_{3,k} & -e_{3,k} \\
\cdot & \cdot & \cdot & \cdot & \cdot & \cdot & \cdot & \cdot \\
0 & \cdot & \cdot & 0 & d_{k-1,k} & e_{k-1,k} & -d_{k-1,k} & -e_{k-1,k}
\end{bmatrix}
$$

Figure A1. The matrix **B**.

B is a matrix of coefficients given in Fig. A1, where

$$d_{i,j}=\frac{(x_i-x_j)}{r_{i,j}} \quad \text{and} \quad e_{i,j}=\frac{(y_i-y_j)}{r_{i,j}}.$$

W is an $m \times m$ element diagonal matrix, with the diagonal elements being given by $w_{i,i}=\frac{1}{\sigma_i^2}$, where σ_i is the expected standard deviation of distance measurement i.

The vector of residuals is calculated from:

$$V=F-\mathbf{B}X, \tag{A2}$$

where:

$$V=[v_{1,2}v_{1,3}\ldots v_{1,k}\, v_{2,3}\ldots v_{2,k}\ldots\ldots v_{k-1,k}]^{\mathrm{T}}$$

and $v_{i,j}$ is the residual for the distance measurement between stations i and j. The unit variance, σ_0^2, is calculated from:

$$\sigma_0^2=\frac{V^{\mathrm{T}}\mathbf{W}V}{m-n}, \tag{A3}$$

and the standard error is equal to the square root of the unit variance.

Estimates of the standard deviations of the calculated X, Y co-ordinates are obtained from the matrix:

$$\mathbf{Q}=\sigma_0^2\,(\mathbf{B}^{\mathrm{T}}\mathbf{W}\mathbf{B})^{-1}. \tag{A4}$$

The square roots of the diagonal elements of **Q** are the required standard deviations and are in the order

$$\sigma_{y_2}, \sigma_{x_3}, \sigma_{y_3}, \sigma_{x_4}, \sigma_{y_4}, \ldots \sigma_{x_k}, \sigma_{y_k}.$$

In the current implementation of the algorith, σ_0^2 has not been included in the calculation of **Q** on the grounds that there is considerable uncertainty in σ_0^2 for surveys in which m–n is small (Cross, 1981), in other words, for surveys in which not many more than the bare minimum number of inter-station distances have been measured.

Appendix B. Least squares adjustment algorithm

The least squares adjustment algorithm consists of the following steps:

1. Use the estimated distance measurement standard deviations to set up the weight matrix, **W**.
2. Use the estimated X, Y co-ordinates of the stations and the measured distances to calculate the elements of matrix **B** and vector F.
3. Calculate vector X using equation (A1).
4. If the magnitude of the largest component of X is less than the desired calculation accuracy, go to step (5), otherwise calculate new estimates of the station co-ordinates from:
   $x_i'=x_i+\delta x_i$,
   $y_i'=y_i+\delta y_i$.
   (Note that x_1, y_1 and x_2 are not adjusted.) Using these new estimates of the station co-ordinates, repeat from step (2).
5. Use equations (A2), (A3) and (A4) to calculate the residuals, standard error and co-ordinate standard deviations.
6. Output results:

Appendix C. The program

```
10    REM PROGRAM LSQADJ
20    REM PROGRAM TO CALCULATE STATION CO-ORDINATES
30    REM GIVEN INTER-STATION DISTANCES.
40    REM USES METHOD OF LEAST SQUARES
50    REM MAXIMUM OF 20 STATIONS WITH CURRENT ARRAY DIMENSIONS
60    REM
70    REM        A.J. DUNCAN    12/12/86
80    REM
90    REM MODIFICATIONS JEREMY GREEN 21/7/87,10/8/87
100   REM INPUT FILE FORMAT:
110   REM NSTAT
120   REM E(1)        N(1)
130   REM  .           .
140   REM  .           .
150   REM E(NSTAT) N(NSTAT)
160   REM R1,2        S1,2
170   REM R1,3        S1,3
180   REM  .           .
190   REM  .           .
200   REM R1,NSTAT  S1,NSTAT
210   REM R2,3        S2,3
220   REM  .           .
230   REM  .           .
240   REM R2,NSTAT  S2,NSTAT
250   REM  .           .
260   REM  .           .
270   REM RNSTAT-1,NSTAT  SNSTAT-1,NSTAT
280   REM
290   REM WHERE:
300   REM         NSTAT = NO. OF STATIONS  (MAX =20)
310   REM         E(I),N(I) = ESTIMATED X,Y, CO-ORDINATES OF STATION I
320   REM         RI,J     = MEASURED DISTANCE BETWEEN STATIONS I AND J
330   REM         SI,J     = ESTIMATED STANDARD DEVIATION OF RI,J
340   REM NOTE:
350   REM       FOR ANY DISTANCES THAT WERE NOT MEASURED, AN APPROXIMATE
360   REM       VALUE SHOULD BE ENTERED ALONG WITH A LARGE STANDARD
370   REM       DEVIATION   (SAY 1000 TIMES ESTIMATED STANDARD DEVIATION OF
380   REM         MEASURED DISTANCES)
390   REM
400   REM   FIRST DIMENSION OF B >= NSTAT(NSTAT-1)/2
410   REM   SECOND DIMENSION OF B >= NDIM =2 *NSTAT-3
420   REM    DIMENSION OF NI >=  NDIM(NDIM+1)/2
430   REM
440    OPTION BASE 1
450      DIM  B(190,37),W(190),F(190),DEL(37),NI(703)
460   DIM  E(20),N(20),R(190),TVEC(37),V(190),XX(37)
470     REM
480   NSTMAX=20
490   REM CONLIM = MAX CHANGE IN CO-ORDINATES FOR CONVERGENCE
500      CONLIM=.001
510   REM NITMAX = NO. OF ITERATIONS ALLOWED
520      NITMAX=50
530      NIT = 0
540     REM ROUTINE LQ1:
550      PRINT
560     REM
570      PRINT " **PROGRAM LSQADJ: LEAST SQUARES ADJUSTMENT OF SURVEY
DATA**"
580      REM
590    REM DATA**"
600      PRINT
610      INPUT "FILE NAME FOR INPUT DATA "; FIN$
620      OPEN FIN$ FOR INPUT AS #1
630      INPUT #1,NSTAT
640      PRINT NSTAT
650      REM
660      IF NSTAT>2 AND NSTAT<=NSTMAX THEN GOTO 700:REM ROUTINE LQ10
670      PRINT "ERROR: NO. STATIONS MUST BE IN RANGE 3, ";NSTMAX
680      CLOSE #1
690      GOTO 540 :REM ROUTINE LQ1
700    REM ROUTINE LQ10:
710     FOR I=1 TO NSTAT
720      INPUT #1,E(I),N(I)
730      PRINT E(I),N(I)
```

```
740     REM
750    NEXT I
760    NN=2*NSTAT-3
770    K=1
780    NTEMP=NSTAT-1
790    FOR I=1 TO NTEMP
800    REM
810     FOR Q=1 TO NTEMP:PRINT,CHR$(9);:NEXTQ
820      ITEMP=I+1
830      FOR J=ITEMP TO NSTAT
840         INPUT #1,R(K),S
850         PRINT USING "#####.###"; R(K),S
860           REM
870        W(K)=1/S^2
880         FOR KK=1 TO NN
890           B(K,KK)=0
900          NEXT KK
910          K=K+1
920      NEXT J
930    NEXT I
940    CLOSE #1
950    END
960    MM=K-1
970    PRINT
980    PRINT
990    INPUT "FILE NAME FOR OUTPUT DATA",FOUT$
1000   OPEN FOUT$ FOR OUTPUT AS #2
1010  REM
1020  REM SET UP MATRICES
1030  REM ROUTINE LQ20:
1040   PRINT "ITERATION NO. ";NIT
1050   REM
1060   K=1
1070   NTEMP=NSTAT-1
1080   FOR I=1 TO NTEMP
1090     ITEMP=I+1
1100     FOR J=ITEMP TO NSTAT
1110      C=SQR((E(I)-E(J))^2+(N(I)-N(J))^2)
1120       F(K)=R(K)-C
1130      AA=(E(I)-E(J))/C
1140      BB=(N(I)-N(J))/C
1150       IA=2*I-4
1160       IF IA > 0 THEN B(K,IA)=AA
1170       IA=IA+1
1180       IF IA > 0 THEN  B(K,IA)=BB
1190       IA=2*J-4
1200       IF IA > 0 THEN B(K,IA)=-1*AA
1210       IA=IA+1
1220      IF IA>0 THEN B(K,IA)=-1*BB
1230       K=K+1
1240     NEXT J
1250   NEXT I
1260   REM
1270    GOSUB 1940 :REM ROUTINE LEASQ
1280          REM
1290   IF IERR=1 THEN GOTO 1880:REM ROUTINE LQERR
1300   REM
1310   REM ADJUST STATION CO-ORDS
1320   N(2)=DEL(1)+N(2)
1330   K=2
1340   FOR I=3 TO NSTAT
1350     E(I)=E(I)+DEL(K)
1360      K=K+1
1370     N(I)=N(I)+DEL(K)
1380      K=K+1
1390   NEXT I
1400   FOR I=1 TO NSTAT
1410      PRINT USING "X = #####.###  Y=#####.###";E(I); N(I)
1420   REM
1430    NEXT I
1440  REM
1450  REM CHECK FOR CONVERGENCE
1460   FOR I=1 TO NN
1470      IF ABS(DEL(I))>CONLIM THEN GOTO 1500:REM ROUTINE LQ30
1480   NEXT I
1490   GOTO 1550:REM ROUTINE LQ40
1500  REM ROUTINE LQ30:
1510   NIT=NIT+1
```

```
1520   IF NIT <=NITMAX THEN GOTO 1030:REM ROUTINE LQ20
1530   PRINT "ERROR: CONVERGENCE FAILURE"
1540   PRINT #2,"ERROR: CONVERGENCE FAILURE"
1550  REM ROUTINE LQ40:
1560   K=-2
1570   PRINT #2,"FIX"+CHR$(9)+"X CO-ORD"+CHR$(9)+"Y CO-ORD"+"STD DEV
X"+CHR$(9)+ "STD DEV Y"
1580   FOR I=1 TO NSTAT
1590     SE=0
1600     SN=0
1610     KSUB=K*(K+1)/2
1620     IF K>0 THEN SE=SQR(NI(KSUB))
1630     K=K+1
1640     KSUB=K*(K+1)/2
1650     IF K>0 THEN SN=SQR(NI(KSUB))
1660     K=K+1
1670     E(I)=INT(E(I)*1000+.5)/1000
1680     N(I)=INT(N(I)*1000+.5)/1000
1690     SE=INT(SE*1000+.5)/1000
1700     SN=INT(SN*1000+.5)/1000
1710     PRINT #2,I;CHR$(9);E(I);CHR$(9);N(I);CHR$(9);SE;CHR$(9);SN
1720   NEXT I
1730  REM CALCULATE RESIDUALS
1740   GOSUB 2510:REM SUBROUTINE RESIDU
1750   SDEV=SQR(SIGSQ)
1760   SDEV=INT(SDEV*1000+.5)/1000:PRINT#2 , "STANDARD ERROR = ";SDEV
1770  REM ROUTINE TO PRINT MATRIX OF RESIDUALS
1780  PRINT#2,"RESIDUALS :":TX=NSTAT-1
1790   FOR Q=1 TO TX
1800                FOR R=1 TO Q : PRINT#2,CHR$(9);:NEXT R
1810         FOR S=1 TO (TX-Q+1)
1820                I=(Q-1)*TX+S:V(I)=INT(V(I)*1000+.5)/1000:PRINT #2,I,V(I)
1830         NEXT S
1840   NEXT Q
1850   CLOSE #2
1860   GOTO LQ1
1870   REM
1880   REM ROUTINE LQERR:
1890     PRINT " ERROR: MATRIX IS SINGULAR"
1900     CLOSE #2
1910     GOTO 540:REM ROUTINE LQ1
1920  REM
1930  REM
1940  REM SUBROUTINE LEASQ:
1950  REM  SUBROUTINE TO PERFORM GENERAL LEAST SQUARES CALCULATION:
1960  REM              DEL=INV(BtWB)BtWF
1970  REM  WHERE:
1980  REM           B IS AN MM * NN MATRIX OF COEFFICIENTS
1990  REM           W IS A DIAGONAL MM * MM MATRIX OF WEIGHTS
2000  REM              ( STORED AS AN MM ELEMENT VECTOR )
2010  REM           F IS  AN MM ELEMENT VECTOR OF CONSTANTS
2020  REM           DEL IS AN NN ELEMENT VECTOR OF PARAMETERS
2030  REM           NI IS AN ARRAY OF LENGTH NN(NN+1)/2 WHICH RETURNS
2040  REM                INV(BtWB) WITH ELEMENTS STORED SO THAT:
2050  REM                ELEMENT(I,J)=NI(I*(I-1)/2+J)
2060  REM           TVEC IS A WORKING VECTOR  OF LENGTH NN
2070  REM
2080  REM                A.J. DUNCAN  12/12/86
2090  REM
2100  REM
2110  REM FORM BtWB
2120     IK=1
2130     FOR I=1 TO NN
2140        FOR J=1 TO I
2150           SUM=0
2160           FOR K=1 TO MM
2170            SUM=SUM+W(K)*B(K,I)*B(K,J)
2180           NEXT K
2190           NI(IK)=SUM
2200           IK=IK+1
2210        NEXT J
2220     NEXT I
2230  REM
2240  REM INVERT MATRIX
2250
2260     GOSUB 2860:REM ROUTINE PDINV
2270
2280     IF IERR=1 THEN RETURN
```

```
2290 REM
2300 REM CALCULATE BtWF
2310   FOR I=1 TO NN
2320     SUM=0
2330     FOR K=1 TO  MM
2340       SUM=SUM+W(K)*B(K,I)*F(K)
2350     NEXT K
2360     TVEC(I)=SUM
2370   NEXT I
2380 REM
2390 REM FORM RESULT VECTOR
2400   FOR I=1 TO NN
2410     SUM=0
2420     FOR K=1 TO NN
2430       IF I >= K THEN ISUB=I*(I-1)/2+K
2440       IF I< K THEN ISUB=K*(K-1)/2+I
2450       SUM=SUM+NI(ISUB)*TVEC(K)
2460     NEXT K
2470     DEL(I)=SUM
2480   NEXT I
2490   RETURN
2500 REM
2510 REM SUBROUTINE RESIDU:
2520 REM SUBROUTINE TO CALCULATE RESIDUALS AFTER LEAST SQUARES
2530 REM ADJUSTMENT HAS BEEN CARRIED OUT.
2540 REM          V = F-B.DEL
2550 REM ALSO CALCULATES UNIT VARIANCE:
2560 REM
2570 REM          SIGSQ = VtWV/(MM-NN)
2580 REM
2590 REM WHERE:
2600 REM     B IS AN MM * NN MATRIX OF COEFFICIENTS
2610 REM     W IS A DIAGONAL MM * MM MATRIX OF WEIGHTS
2620 REM       (STORED AS AN MM ELEMENT VECTOR)
2630 REM     F IS AN M ELEMENT VECTOR OF CONSTANTS
2640 REM     DEL IS AN NN ELEMENT VECTOR OF PARAMETERS
2650 REM     V IS THE CALCULATED RESIDUAL VECTOR, LENGTH MM
2660 REM
2670 REM          A.J. DUNCAN  12/12/86
2680 REM
2690 SIGSQ=0
2700 FOR I=1 TO MM
2710   SUM1=0
2720   FOR J=1 TO NN
2730     SUM1=SUM1+B(I,J)*DEL(J)
2740   NEXT J
2750   V(I)=F(I)-SUM1
2760   SIGSQ=SIGSQ+W(I)*V(I)^2
2770 NEXT I
2780 IR=MM-NN
2790 IF IR < =0 THEN GOTO 2820:REM ROUTINE RES10
2800 SIGSQ=SIGSQ/IR
2810 RETURN
2820 REM ROUTINE RES10:
2830   SIGSQ=99.99
2840   RETURN
2850 REM
2860 REM SUBROUTINE PDINV:
2870 REM SUBROUTINE TO CALCULATE INVERSE OF A POSITIVE DEFINITE
2880 REM SYMMETRIC MATRIX USING THE BAUER-REINICH ALGORITHM
2890 REM REF: J.C. NASH, "COMPACT NUMERICAL METHODS FOR COMPUTERS:
2900 REM LINEAR ALGEBRA AND FUNCTION MINIMISATION", ADAM HILGER, 1979.
2910 REM
2920 REM  ARRAY NI CONTAINS LOWER TRIANGULAR  ELEMENTS OF ARRAY OF
2930 REM  ORDER N. ELEMENTS ARE STORED SO THAT:
2940 REM       NI(I*(I-1)/2+J) = ARRAY(I,J)
2950 REM  LOWER TRIANGULAR ELEMENTS OF INVERSE ARE RETURNED IN NI
2960 REM
2970 REM
2980 REM  TVEC IS AN N ELEMENT WORKING VECTOR
2990 REM  NMDIM IS THE DIMENSION OF ARRAY NI
3000 REM
3010 REM  IERR = 0 FOR NORMAL RETURN
3020 REM  IERR = 1 IF MATRIX SINGULAR
3030 REM
3040 REM       *** MACINTOSH BASIC VERSION ***
3050 REM
3060 REM       A.J. DUNCAN   12 DECEMBER 1986
```

```
3070 REM
3080 IERR=0
3090 FOR KK=1 TO NN
3100  K=NN-KK+1
3110  S=NI(1)
3120  IF S>0 THEN GOTO PD10
3130  IERR=1
3140  RETURN
3150  PD10:
3160   MMM=1
3170   FOR I=2 TO NN
3180    IQ=MMM
3190    MMM=MMM+I
3200    ISUB1=IQ+1
3210    T=NI(ISUB1)
3220    TVEC(I)=-1*T/S
3230    IF I>K THEN TVEC(I)=-1*TVEC(I)
3240    ITEMP=IQ+2
3250    FOR J=ITEMP TO MMM
3260     ISUB1=J-I
3270     ISUB2=J-IQ
3280     NI(ISUB1)=NI(J)+T*TVEC(ISUB2)
3290    NEXT J
3300   NEXT I
3310   IQ=IQ-1
3320 NI(MMM)=1/S
3330  FOR I=2 TO NN
3340    ISUB1=IQ+I
3350   NI(ISUB1)=TVEC(I)
3360  NEXT I
3370 NEXT KK
3380 RETURN
3390 END
```

References

Cross, P. A., 1981, The computation of position at sea. *Hydrographic Journal,* **20.**

Cross, P. A. & Walker, A. S., 1984, *Statistical and computational aspects of multiposition line fixing.* Symposium on Multi-Position Line Fixing, HMS *Drake,* Plymouth, UK.

Mikhail, M., 1976, *Observations and Least Squares.* IEP.